# 2
# Basics

This chapter introduces the basic capabilities of *Mathematica*, which include simple arithmetic, handling algebraic and trigonometric expressions and assigning values to variables. We will also look at dynamic objects, which allow us to see changes in the variables as they happen.

In this chapter we give a quick introduction to the very basic things one can do with Wolfram *Mathematica*®. We let the reader learn from reading the codes and avoid long and exhausting explanations, as the codes will speak for themselves.

## 2.1 *Mathematica* as a calculator

*Mathematica* can be used as a powerful calculator with the basic arithmetic operations; $+, -$ for addition and subtraction, $*, /$ for multiplication and division and ^ for powers.

```
10^9 - 987654321
12345679

2682440^4 + 15365639^4 + 18796760^4
180630077292169281088848499041

20615673^4
180630077292169281088848499041
```

The last two calculations show that

$$2682440^4 + 15365639^4 + 18796760^4 = 20615673^4,$$



R. Hazrat, *Mathematica®: A Problem-Centered Approach*, Springer Undergraduate Mathematics Series, DOI 10.1007/978-3-319-27585-7_2

disproving a conjecture of Euler that three fourth powers can never sum to a fourth power.[1]

*Mathematica* can handle large calculations:

```
2^9941-1
```

```
346088282490851215242960395767413316722628668900238547790489283445006220809834
```

```
346088282490851215242960395767413316722628668900238547790489283445006220809834
114464364375544153707533664486747635050186414707093323739706083766904042292657
```

If a number of the form $2^n - 1$ happens to be prime, it is called a Mersenne prime. Recall that a *prime number* is a number greater than 1 which is divisible only by 1 and itself. It is easy to see that $2^2 - 1$, $2^3 - 1$ and $2^5 - 1$ are Mersenne primes. The list continues. In 1963, Gillies found that the above number, $2^{9941} - 1$, is a Mersenne prime. With my laptop it takes about 3 seconds for *Mathematica* to check that this is indeed a prime number.[2]

```
PrimeQ[2^9941-1]
True
```

Back to easier calculations:

```
24/17
24/17
```

*Mathematica* always tries to give a precise value, thus it returns back $\frac{24}{17}$ instead of attempting to evaluate the fraction.

[1] This conjecture remained open for almost 200 years, until Noam Elkies at Harvard came up with the above counterexample in 1988.

[2] The largest Mersenne prime found so far is $2^{57885161} - 1$ which was discovered in Jan 2013 and has $17,425,170$ digits.

```
Sin[Pi/5]
```

$\frac{1}{2}\sqrt{\frac{1}{2}(5-\sqrt{5})}$

*Mathematica* gives $\frac{1}{2}\sqrt{\frac{1}{2}(5-\sqrt{5})}$ as the value of $\sin(\pi/5)$, which is the precise value. This shows that *Mathematica* is not approaching the expressions numerically.

In order to get the numerical value, one can use the function `N`.

```
N[24/17]
1.41176

N[24/17, 20]
1.4117647058823529412

?N
N[expr] gives the numerical value of expr. N[expr, n] attempts to
give a result with n-digit precision.
```

As the above line shows, in order to get a quick description of a command, use `?Command`. If you need further explanation use `??Command`. If you need even more, use *Mathematica*'s Document Center in the Help Menu and search for the command. *Mathematica* comes with an excellent Help which contains explanations and many nice examples demonstrating how to use each of the functions available in this software.

All elementary mathematical functions are available here, `Log, Exp, Sqrt, Sin, Cos, Tan, ArcSin,...`. Here we evaluate

$$\sqrt{(\frac{\pi}{4})^2 + (0.5\,\mathrm{Log}[2])^2},$$

```
Sqrt[(Pi/4)^2 + (0.5 Log[2])^2]
0.858466
```

We know that $\sin^2(x) + \cos^2(x) = 1$. We check this using *Mathematica*.

```
Sin[x]^2 + Cos[x]^2
Cos[x]^2 + Sin[x]^2
```

Apparently *Mathematica* has not recognised the identity. We try `Simplify` and `TrigExpand` to see if we can simplify the expression.

```
?Simplify
Simplify[expr] performs a sequence of algebraic and other
transformations on expr, and returns the simplest form it finds.

Simplify[Sin[x]^2 + Cos[x]^2]
1
```

```
?TrigExpand
TrigExpand[expr] expands out trigonometric functions in expr. >>

TrigExpand[Sin[x]^2 + Cos[x]^2]
1
```

This example introduces the commands `Simplify` and `FullSimplify`. If one is not happy with the result, one can always use `Simplify` and even `FullSimplify` to have *Mathematica* work a bit harder to come up with more simplification.

```
?FullSimplify
FullSimplify[expr] tries a wide range of transformations on expr
involving elementary and special functions, and returns the
simplest form it finds.
```

## Problem 2.1

Show that

$$\frac{(1+\sqrt{5})^{10} - (1-\sqrt{5})^{10}}{1024\sqrt{5}} = 55.$$

$\Longrightarrow$ Solution.

If we just enter the expression on the left-hand side of the equality into *Mathematica*, we will not get the result we are after (try it!). Thus we will use `Simplify` to get a more polished answer.

```
Simplify[((1 + Sqrt[5])^10 - (1 - Sqrt[5])^10)/(1024 Sqrt[5])]
55
```

The following problem shows that sometimes one needs to *investigate* a bit more in order to get the desired format of the expressions.

## Problem 2.2

Use *Mathematica* to show that

$$\tan\frac{3\pi}{11} + 4\sin\frac{2\pi}{11} = \sqrt{11}.$$

$\Longrightarrow$ Solution.

We enter the expression into *Mathematica*.

```
Tan[3 Pi / 11] + 4 Sin[2 Pi / 11]
```

$$\mathrm{Cot}\left[\frac{5\,\pi}{22}\right]+4\,\mathrm{Sin}\left[\frac{2\,\pi}{11}\right]$$

Clearly we didn't get $\sqrt{11}$ as an answer. We ask *Mathematica* to try a bit harder by using `Simplify`.

```
Simplify[Tan[3 Pi / 11] + 4 Sin[2 Pi / 11]]
```

$$\mathrm{Cot}\left[\frac{5\,\pi}{22}\right]+4\,\mathrm{Sin}\left[\frac{2\,\pi}{11}\right]$$

Since we didn't get what we were looking for, we try `TrigExpand`.

```
TrigExpand[Tan[3 Pi / 11] + 4 Sin[2 Pi / 11]]
```

$$\frac{2\,\mathrm{Cos}[\frac{\pi}{22}]}{5\,\mathrm{Cos}[\frac{\pi}{22}]^4\,\mathrm{Sin}[\frac{\pi}{22}]-10\,\mathrm{Cos}[\frac{\pi}{22}]^2\,\mathrm{Sin}[\frac{\pi}{22}]^3+\mathrm{Sin}[\frac{\pi}{22}]^5}+$$
$$\frac{\mathrm{Cos}[\frac{\pi}{22}]^5}{5\,\mathrm{Cos}[\frac{\pi}{22}]^4\,\mathrm{Sin}[\frac{\pi}{22}]-10\,\mathrm{Cos}[\frac{\pi}{22}]^2\,\mathrm{Sin}[\frac{\pi}{22}]^3+\mathrm{Sin}[\frac{\pi}{22}]^5}-$$
$$\frac{2\,\mathrm{Cos}[\frac{\pi}{22}]^9}{5\,\mathrm{Cos}[\frac{\pi}{22}]^4\,\mathrm{Sin}[\frac{\pi}{22}]-10\,\mathrm{Cos}[\frac{\pi}{22}]^2\,\mathrm{Sin}[\frac{\pi}{22}]^3+\mathrm{Sin}[\frac{\pi}{22}]^5}-$$
$$\frac{10\,\mathrm{Cos}[\frac{\pi}{22}]^3\,\mathrm{Sin}[\frac{\pi}{22}]^2}{5\,\mathrm{Cos}[\frac{\pi}{22}]^4\,\mathrm{Sin}[\frac{\pi}{22}]-10\,\mathrm{Cos}[\frac{\pi}{22}]^2\,\mathrm{Sin}[\frac{\pi}{22}]^3+\mathrm{Sin}[\frac{\pi}{22}]^5}+$$
$$\frac{72\,\mathrm{Cos}[\frac{\pi}{22}]^7\,\mathrm{Sin}[\frac{\pi}{22}]^2}{5\,\mathrm{Cos}[\frac{\pi}{22}]^4\,\mathrm{Sin}[\frac{\pi}{22}]-10\,\mathrm{Cos}[\frac{\pi}{22}]^2\,\mathrm{Sin}[\frac{\pi}{22}]^3+\mathrm{Sin}[\frac{\pi}{22}]^5}+$$
$$\frac{5\,\mathrm{Cos}[\frac{\pi}{22}]\,\mathrm{Sin}[\frac{\pi}{22}]^4}{5\,\mathrm{Cos}[\frac{\pi}{22}]^4\,\mathrm{Sin}[\frac{\pi}{22}]-10\,\mathrm{Cos}[\frac{\pi}{22}]^2\,\mathrm{Sin}[\frac{\pi}{22}]^3+\mathrm{Sin}[\frac{\pi}{22}]^5}-$$
$$\frac{252\,\mathrm{Cos}[\frac{\pi}{22}]^5\,\mathrm{Sin}[\frac{\pi}{22}]^4}{5\,\mathrm{Cos}[\frac{\pi}{22}]^4\,\mathrm{Sin}[\frac{\pi}{22}]-10\,\mathrm{Cos}[\frac{\pi}{22}]^2\,\mathrm{Sin}[\frac{\pi}{22}]^3+\mathrm{Sin}[\frac{\pi}{22}]^5}+$$
$$\frac{168\,\mathrm{Cos}[\frac{\pi}{22}]^3\,\mathrm{Sin}[\frac{\pi}{22}]^6}{5\,\mathrm{Cos}[\frac{\pi}{22}]^4\,\mathrm{Sin}[\frac{\pi}{22}]-10\,\mathrm{Cos}[\frac{\pi}{22}]^2\,\mathrm{Sin}[\frac{\pi}{22}]^3+\mathrm{Sin}[\frac{\pi}{22}]^5}-$$
$$\frac{18\,\mathrm{Cos}[\frac{\pi}{22}]\,\mathrm{Sin}[\frac{\pi}{22}]^8}{5\,\mathrm{Cos}[\frac{\pi}{22}]^4\,\mathrm{Sin}[\frac{\pi}{22}]-10\,\mathrm{Cos}[\frac{\pi}{22}]^2\,\mathrm{Sin}[\frac{\pi}{22}]^3+\mathrm{Sin}[\frac{\pi}{22}]^5}$$

We try to simplify this expression yet again.

```
Simplify[TrigExpand[Tan[3 Pi / 11] + 4 Sin[2 Pi / 11]]]
```

$$\frac{16\left(i+2\,(-1)^{3/22}-2\,(-1)^{7/22}+2\,(-1)^{15/22}+2\,(-1)^{17/22}+(-1)^{21/22}\right)}{\left(-1+(-1)^{1/11}\right)^5\left(1-10\,\mathrm{Cot}\left[\frac{\pi}{22}\right]^2+5\,\mathrm{Cot}\left[\frac{\pi}{22}\right]^4\right)}$$

Although it is an improvement, we have not yet got $\sqrt{11}$. We try again, this time with `FullSimplify`.

```
FullSimplify[TrigExpand[Tan[3 Pi / 11] + 4 Sin[2 Pi / 11]]]
```

$\sqrt{11}$

We thus obtained the result we were after. We could also have asked *Mathematica* directly whether the right-hand side of the identity is the same as the left-hand side.

```
Tan[3 Pi / 11] + 4 Sin[2 Pi / 11] == Sqrt[11]
```

$\mathrm{Cot}\left[\frac{5\pi}{22}\right]+4\,\mathrm{Sin}\left[\frac{2\pi}{11}\right]==\sqrt{11}$

```
FullSimplify[Tan[3 Pi / 11] + 4 Sin[2 Pi / 11] == Sqrt[11]]
```

```
True
```

For a complete list of elementary functions have a look at Functional Navigator: Mathematics and Algorithms: Mathematical Functions in *Mathematica*'s Help.

### *Exercise 2.1*

Show that $\sqrt{\sqrt[3]{64}\left(2^2+(1/2)^2\right)-1}=4$.

### *Exercise 2.2*

Show that

$$\left(\frac{1}{2}+\cos\left(\frac{\pi}{20}\right)\right)\left(\frac{1}{2}+\cos\left(\frac{3\pi}{20}\right)\right)\left(\frac{1}{2}+\cos\left(\frac{9\pi}{20}\right)\right)\left(\frac{1}{2}+\cos\left(\frac{27\pi}{20}\right)\right)=\frac{1}{16}.$$

## Problem 2.3

Using *Mathematica*, explain why $4 + 6/4 * 3\hat{} - 2 + 1 = \dfrac{31}{6}$.

$\Longrightarrow$ Solution.

If you look at *Mathematica*'s Help, under "Special Ways to Input Expressions", you will see the following note: "The *Mathematica* language has a definite grammar which specifies how your input should be converted to internal form." One aspect of the grammar is that it specifies how pieces of your input should be grouped. The general rule is that if $\otimes$ has higher precedence than $\oplus$, then $a \oplus b \otimes c$ is interpreted as $a \oplus (b \otimes c)$, and $a \otimes b \oplus c$ is interpreted as $(a \otimes b) \oplus c$. You will then find a long table listing which operation has a higher precedence and thus, based on that, you will be able to explain why $4 + 6/4 * 3\hat{} - 2 + 1$ amounts to $\frac{31}{6}$.

However, common sense tells us that instead of creating an ambiguous expression such as $4 + 6/4 * 3\hat{} - 2 + 1$, one should use parentheses () to group objects together and make the expression more clear. For example, one could write $4 + \Big((6/4) * 3\hat{}(-2)\Big) + 1$, or even better use *Mathematica*'s Palette (Basic Math Assistance) and type

$$4 + \frac{6}{4} \times 3^{-2} + 1.$$

## ♣ TIPS

- The mathematical constant $e$, the exponential number, is defined in *Mathematica* as `E`, or `Exp`. To get $e^n$ use either `E^n` or `Exp[n]`. The constant $\pi$ can be typed as `Pi`.

- Comments can be added to the codes using `(* comment *)`.

```
(* the most beautiful theorem in Mathematics *)
  E^(I Pi) + 1
  0
```

- The symbol `%` refers to the previous output produced. `%%` refers to the second previous output, and so on.

- If in calculations you don't get what you are expecting, use `Simplify` or even `FullSimplify` (see Problem 2.2).

- To get a numerical approximation, use `N[expr]` or alternatively, `expr//N` (see Problem 3.3 for different ways of applying a function to a variable).

Use `EngineeringForm[expr,n]` and `ScientificForm[expr,n]` to get other forms of numerical approximations to $n$ significant digits.

## 2.2 Numbers

There are several standard ways to start with an integer and produce new numbers out of it. For example, starting from 4, one can form $4 \times 3 \times 2 \times 1$, which is represented by 4!.

```
4!
24

123!
1214630436702532967576624324188129585545421708848338231532891
8161829235892362167668831156960612640202170735835221294047782
5910915704116514721860295199062616467307339074198149529600 0
0000000000000000000000000
```

The fundamental theorem of arithmetic states that one can decompose any natural number $n$ as a product of powers of primes and this decomposition is unique, i.e., $n = p_1^{k_1} \cdots p_t^{k_t}$ where $p_i$'s are prime. Thus $12 = 2^2 \times 3^1$ and $37534 = 2 \times 7^2 \times 383$. *Mathematica* can do all of these:

```
FactorInteger[12]
{{2, 2}, {3, 1}}

FactorInteger[37534]
{{2,1},{7,2},{383,1}}

2^1 * 7^2 * 383^1
37534

FactorInteger[6473434456376432]
{{2,4},{3239053,1},{124909859,1}}

PrimeQ[124909859]
True

Prime[8]
19
```

`Prime[n]` produces the $n$-th prime number. `PrimeQ[n]` determines whether $n$ is a prime number. More than 2200 years ago Euclid proved that the set of prime numbers is infinite. His proof is used even today in modern books. However, it is not that long ago that we also learned that there is no simple formula that produces only prime numbers.

In 1640 Fermat conjectured that the formula $2^{2^n}+1$ always produces a prime number. Almost a hundred years later the first counterexample was found.

```
PrimeQ[2^(2^1)+1]
True

PrimeQ[2^(2^2)+1]
True

PrimeQ[2^(2^3)+1]
True

PrimeQ[2^(2^4)+1]
True

PrimeQ[2^(2^5)+1]
False

2^(2^5)+1
4294967297

FactorInteger[2^(2^5)+1]
{{641,1},{6700417,1}}
```

This shows that $2^{2^5}+1$ is not a prime number. In fact it decomposes into two prime numbers $2^{2^5}+1 = 641 \times 7600417$.

## Problem 2.4

What is the probability that a randomly chosen 13-digit number will be a prime?

$\Longrightarrow$ SOLUTION.

The probability is the number of 12-digit prime numbers over the number of all 12-digit numbers. So we start by finding how many 12-digit numbers exist:

```
10^13 - 10^12
9000000000000
```

Next, we will find how many 13-digit prime numbers exist. We will use the following built-in function of *Mathematica*.

```
?PrimePi

PrimePi[x] gives the number of primes less than
or equal to x. >>

PrimePi[10^13]
346065536839

PrimePi[10^12]
37607912018

N[(346065536839 - 37607912018)/9000000000000]*100
3.42731
```

So the probability that we randomly pick a 12-digit prime number is only 3.42 percent.

*Exercise 2.3*

Check that 12345678909876543211 is a prime number.

*Exercise 2.4*

Check that for any given positive integer $n \geq 3$, the least common multiple of the numbers $1, 2, \cdots, n$ is greater than $2^{n-1}$. (Hint: see `LCM`.)

*Exercise 2.5*

Check that the number 32! ends with 7 zeros.

## Problem 2.5

The function `Mod[m,n]` gives the remainder on division of $m$ by $n$. Let $a = 98$ and $b = 75$. Show that $a^a b^b$ ends with exactly 98 zeros.

$\Longrightarrow$ SOLUTION.

Clearly you can directly compute $a^a b^b$ and count the number of zeros as in Exercise 2.5. But this time you have to count whether the number ends with 98 zeros. Here is a more clever approach: Check whether the remainder of this number when divided by $10^{98}$ is zero, but its remainder on division by $10^{99}$ is not.

```
Mod[98^98 * 75^75, 10^98]
0

Mod[98^98 * 75^75, 10^99]
5000000000000000000000000000000000000000000000000000000000000
00000000000000000000000000000000000000
```

## Problem 2.6

Recall that for integers $m$ and $n$, the binomial coefficient $\binom{n}{m}$ is defined as $\dfrac{n!}{m!(n-m)!}$. Using *Mathematica*, check that

$$\binom{m+n}{m} = \frac{(m+n)!}{m!n!}.$$

⟹ SOLUTION.

Not only is the binomial function available in *Mathematica*, but *Mathematica* can also perfectly handle it symbolically, as the solution to this problem shows. We will talk more about symbolic computations in Section 2.3.

```
Binomial[m + n, n] == (n + m)!/(n! m!)
Binomial[m + n, n] == (m + n)!/(m! n!)

FullSimplify[Binomial[m + n, n] == (n + m)!/(n! m!)]
True
```

This is another instance where we need to use `FullSimplify` to make *Mathematica* work harder to come up with the result.

We will discuss the different equalities available in *Mathematica* in Section 2.6. However, for the time being, note that `==` is used to compare both sides of equations.

There are several more integer functions available in *Mathematica*, which can be found in Functional Navigator: Mathematics and Algorithms: Mathematical Functions: Integer Functions.

## ♣ TIPS

– The command `NextPrime[n]` gives the next prime larger than `n` and `PrimePi[n]` gives the number of primes less than or equal to `n` (see Problem 2.4).

– For integers $m$ and $n$, one can find unique numbers $q$ and $r$ such that $r$ is positive, $m = qn + r$ and $r <| q |$. Then `Mod[m,n]=r` and `Quotient[m,n]=q`.

– If an evaluation is taking a long time, in order to stop the evaluation use Alt+. (for Windows) and Cmd+. (for Apple Macintosh). For example, try to calculate the 1234567891011-th prime number. If you can't wait to get the result, you now know how to stop the process. There are cases where pressing Alt+. does not help, even if you do it several times. In these situations, use the Evaluation menu and choose Quit Kernel.

## 2.3 Algebraic computations

One of the abilities of *Mathematica* is to handle symbolic computations, i.e., *Mathematica* can comfortably work with symbols (we have seen one example of this in Problem 2.6). Consider the expression $(x+1)^2$. One can use *Mathematica* to expand this expression:

```
Expand[(1 + x)^2]
1 + 2 x + x^2
```

*Mathematica* can also do the inverse of this task, namely factorise an expression:

```
Factor[1 + 2 x + x^2]
(1 + x)^2
```

While expansion of an algebraic expression is a simple and routine procedure, the factorization of algebraic expressions is often quite challenging. My favorite example is this one. Try to factorise the expression $x^{10}+x^5+1$. Here is one way to do it:

$$
\begin{aligned}
&x^{10}+x^5+1 \quad \text{(adding } x^i - x^i, 1 \leq i \leq 9 \text{, to the expression we have)}\\
&= x^{10} + \underbrace{x^9 - x^9} + \underbrace{x^8 - x^8} + \cdots + \underbrace{x^6 - x^6} +\\
&\underbrace{+x^5 - x^5} + x^5 + \underbrace{x^4 - x^4} + \cdots + \underbrace{x - x} + 1 \quad \text{(now rearranging the terms)}\\
&= x^{10} + x^9 + x^8 - x^9 - x^8 - x^7 + x^7 + x^6 + x^5 - x^6 - x^5 - x^4\\
&\qquad + x^5 + x^4 + x^3 - x^3 - x^2 - x + x^2 + x + 1\\
&= x^8(x^2+x+1) - x^7(x^2+x+1) + x^5(x^2+x+1) - x^4(x^2+x+1)\\
&\qquad + x^3(x^2+x+1) - x(x^2+x+1) + x^2+x+1\\
&= (x^2+x+1)(x^8 - x^7 + x^5 - x^4 + x^3 - x + 1)
\end{aligned}
$$

*Mathematica* can easily come up with this factorization:

```
Factor[x^10 + x^5 + 1]
(1 + x + x^2) (1 - x + x^3 - x^4 + x^5 - x^7 + x^8)
```

## Problem 2.7

Prove that the product of four consecutive numbers plus one is always a square number.

⟹ SOLUTION.

We first check this is indeed the case for an example:

```
Sqrt[13*14*15*16 + 1]
209
```

Now here is a proof:

```
Factor[n (n + 1) (n + 2) (n + 3) + 1]
(1 + 3 n + n^2)^2
```

## Problem 2.8

Here is a formula to generate many instances where the sum of three fourth powers is a fourth power. Check this with *Mathematica* and find some examples of such numbers.

$$(85v^2 + 484v - 313)^4 + (68v^2 - 586v + 10)^4 + (2u)^4 = (357v^2 - 204v + 363)^4,$$

where

$$u^2 = 22030 + 28849v - 56158v^2 + 36941v^3 - 31790v^4.$$

$\Longrightarrow$ SOLUTION.

This solution is not optimal. We give another solution to this in Exercise 10.2 in Chapter 10 based on the pattern matching approach.

We use `Sqrt` to write $u$ in terms of $v$ and then enter the formula into *Mathematica*, asking it to compare the two sides of the equality using `==`.

```
u = Sqrt[22030 + 28849 v - 56158 v^2 +  36941 v^3 - 31790 v^4];

(85 v^2 + 484 v - 313)^4 + (68 v^2 - 586 v + 10)^4 + (2 u)^4 ==
(357 v^2 - 204 v + 363)^4

(10 - 586 v + 68 v^2)^4 + (-313 + 484 v + 85 v^2)^4 +
16 (22030 + 28849 v - 56158 v^2 + 36941 v^3 -
     31790 v^4)^2 == (363 - 204 v + 357 v^2)^4
```

Since we didn't get an answer, we use `Simplify` to make *Mathematica* work harder.

```
Simplify[(85 v^2 + 484 v - 313)^4 +
 (68 v^2 - 586 v + 10)^4 + (2 u)^4 == (357 v^2 - 204 v + 363)^4]
True
```

# ♣ TIPS

– The command `Together` converts a sum of terms into a single term over a common denominator. The command `Apart` (almost) does the reverse of `Together` (see Exercise 2.7).

*Exercise 2.6*

Factorise the polynomial $(1+x)^{30} + (1-x)^{30}$.

*Exercise 2.7*

Using `Together`, write the expression

$$\frac{1}{1+x} + \frac{1}{1+\frac{1}{1+x}}$$

with a single denominator. Now apply `Apart` to the result to get an expression as a sum of terms with minimal denominators.

There are several more algebraic functions available in *Mathematica*, which can be found in Functional Navigator: Mathematics and Algorithms: Mathematical Functions: Polynomial Algebra.

## 2.4 Trigonometric computations

Similar to algebraic expressions (Section 2.3), *Mathematica* can handle trigonometric expressions, as we saw in Problem 2.2. Here one uses `TrigExpand` and `TrigFactor` to work with trig. expressions.

*Mathematica* is quite at ease with trig. identities, as the following problem demonstrates.

### Problem 2.9

Using *Mathematica*, check that the following trigonometric identities hold:

$$\sin^3(x)\cos^3(x) = \frac{3\sin(2x) - \sin(6x)}{32}$$

$$\frac{1+\sin(x)-\cos(x)}{1+\sin(x)+\cos(x)} = \tan(x/2)$$

$\Longrightarrow$ SOLUTION.

The only challenge here is to translate these expressions correctly into *Mathematica*.

```
Simplify[Sin[x]^3 Cos[x]^3 == (3 Sin[2 x] - Sin[6 x])/32]
True

Simplify[(1 + Sin[x] - Cos[x])/(1 + Sin[x] + Cos[x]) == Tan[x/2]]
True
```

Note that == is used to compare both sides of equations. We will discuss the different equalities available in *Mathematica* in Section 2.6.

*Exercise 2.8*

Using *Mathematica*, show that

$$\frac{1+\sin(x)-\cos(x)}{1+\sin(x)+\cos(x)} = \tan(x/2).$$

## ♣ TIPS

– The argument of trig. functions, e.g., `Sin`, is assumed to be in radians. (Multiply by `Degree` to convert from degrees to radians.)

```
Sin[30 Degree]
1/2
```

## 2.5 Variables

In order to feed data into a computer program one needs to define variables to be able to assign data to them. As long as you use common sense, any names you choose for variables are valid in *Mathematica*. Names like `x, y, x3, myfunc, xQuaternion,...` are all fine. Do not use an underscore _ to define a variable.[3] Also note that *Mathematica* is case sensitive, thus `xy` and `xY` are considered as two different variables.

```
x = 3
3

y = 4
4

x^2 + y^2
25

Sqrt[x^2 + y^2]
5
```

If we need to enter several statements in one line, we can separate them with `;`.

[3] This is quite common in Pascal or C, to define variables such as `x_printer, com_graph,...`. In *Mathematica*, the underscore is reserved and will be used in the definition of functions in Chapter 3.

```
t = 7; s = 4; t!/(s! (t - s)!)
35
```

One can assign any type of data to a variable. Not only would *Mathematica* not complain, but she would also carry out all the computations with that specific data. Here we drag a picture to the *Mathematica* front end (editor) and assign it to the variable x.

```
x =
```


```
x/(1 + x/(1 + x))
```

## Problem 2.10

Using `Expand`, for `Expand[(1+x)^ 2]`, instead of obtaining $1 + 2x + x^2$ we get

```
Expand[(1 + x)^2]
16
```

What seems to be the problem?

⟹ SOLUTION.

If you are working through this section, at the beginning of this section you will already have defined `x=3`. Thus *Mathematica* will take this into account when working with the expression $(1+x)^2$, which then amounts to 16. This demonstrates one of the common mistakes one tends to make in *Mathematica*, namely using variables which have already been defined, as undefined symbols. In order to clear the value or definition of a variable, use `Clear`.

```
Clear[x]

Expand[(1 + x)^2]
1 + 2 x + x^2
```

## ♣ TIPS

- Use `Clear[x]` to clear the value given to the variable `x`, before using `x` as a symbol.
- Use `Clear["Global‘*"]` to clear values and definitions given to *all the* symbols.
- Assigning a value to a symbol works globally. That means, if you open a new NoteBook, the values given to variables in a previous NoteBook still exist.

## 2.6 Equalities =, :=, ==

Primarily there are three equalities in *Mathematica*, =, := and ==. There is a fundamental difference between = and :=. Study the following example:

```
x=5;y=x+2;

y
7

x=10
10

y
7

x=15
15

y
7
```

So changing the value of `x` does not affect the value of `y`. Now compare this with the following example, where we replace = with := in the definition of `y`.

```
x=5;y:=x+2;

y
7

x=10
10

y
12

x=15
15

y
17
```

From the first example it is clear that when we define y=x+2 then y takes the *value* of x+2 and this will be assigned to y. No matter if x changes its value, the value of y remains the same. In other words, y is independent of x. But in y:=x+2, y is dependent on x, and when x changes, the value of y changes too. That is using := makes y a function with variable x. The following is an excellent example demonstrating the difference between = and :=.

```
?Random

Random[ ] gives a uniformly distributed pseudorandom Real in the
range 0 to 1.

x=Random[]
0.246748

x
0.246748

x
0.246748

x:=Random[]

x
0.60373

x
0.289076

x
0.564378
```

When defining `x=Random[]`, the function `Random` generates a number and this number will be assigned to `x`. Each time we call on `x`, this number is what we get. However, when we define `x:=Random[]`, then the definition of `x` is

`Random[]`. Thus when we call `x`, we have in fact called on `Random` which then generates a new random number.

We will examine this difference between `=` and `:=` again in Example 4.7.

Finally, the equality == is used to compare:

```
5==5
True

3==5
False
```

We will discuss this further in Section 6.1.

## 2.7 Dynamic variables

The new version of *Mathematica*[4] comes with an ability to define *dynamic* variables. This means one can monitor the changes in a variable "live", i.e., as they happen. We are going to introduce this feature early in the book to take advantage of it as we go along.

We saw in Section 2.5 that one can define variables and assign values to them.

```
x = 3
3

x = 10
10
```

Here when we assign 10 to `x`, although this is the new value of `x`, in the line above it, i.e., `x=3, 3` does not change. However, if we define the variable `x` as a dynamic variable, then each time we change the value of `x` anywhere in the program, all the old values also change to the new value accordingly.

```
Dynamic[x]
10
```

Then if in the next line we change the value to `x=15`, we will see that the value of the previous line immediately changes to 15 as well.

```
Dynamic[x]
15

x=15
15
```

One can control the value of the variable `x` by introducing a *slider*.

[4] Currently version 10.

```
Slider[Dynamic[x]]
```

```
Dynamic[x]
```

0.326

You will see that as you drag the slider, the value of `x` changes. This already gives us a lot of power, as the following example will show. Recall from Section 2.3 that we can expand expressions using `Expand`.

```
Expand[(1 + y)^2]
1 + 2 y + y^2

Expand[(1 + y)^3]
1 + 3 y + 3 y^2 + y^3
```

Now we can simply consider $(1+y)^n$ and then, defining $n$ as a dynamic variable and controlling it with a slider, we can change the value of $n$ by dragging the slider and see the expansions of $(1+y)^n$ for different values of $n$ as they happen right in front of our eyes!

```
Slider[Dynamic[n], {1, 10, 1}]
```

```
Dynamic[Expand[(1 + y) ^n]]
```

$1 + 7y + 21y^2 + 35y^3 + 35y^4 + 21y^5 + 7y^6 + y^7$

Note that, when defining `Slider`, the value of $x$ varies from 0 to 1. If we want to change this interval, as in the previous example, we can specify the interval and the step that is added to $x$ each time by using {`xmin,xmax, step`}.

A similar concept to `Slider` is the function `Manipulate` which allows us to change the value of a variable and see the result "live".

```
Manipulate[Expand[(1 + y) ^n], {n, 1, 10, 1}]
```



The control buttons can be used to start or stop the process, and make it faster or slower. Just try it out.

We will see later, for example in Chapter 14 when we deal with graphics, that we can use `Manipulate` to change the value of our parameters and see how the graph changes accordingly.

## Problem 2.11

Using `Manipulate`, observe that the polynomial $x^{2n}+x^n+1$ can be decomposed into smaller factors for any $1 \le n \le 20$ except $n = 1, 3, 9$.

⟹ Solution.

```
Manipulate[Factor[x^(2 n) + x^n + 1], {n, 1, 20, 1}]
```




## Problem 2.12

Using `Manipulate`, find out for which positive integers $n$ and $m$, between 1 and 100, $m^2 + n^2$ is a square number (these are called Pythagorean pairs).

⟹ Solution.

Later in Chapter 8, when we are discussing loops, we will write a program to generate these numbers (Problem 8.12). Here we will use `Manipulate`, defining two dynamic variables $m$ and $n$, and we will look at the result of $\sqrt{m^2+n^2}$. If this is an integer, then $(m, n)$ is a Pythagorean pair.

```
Manipulate[Sqrt[m^2 + n^2], {m, 1, 100, 1}, {n, 1, 100, 1}]
```



http://www.springer.com/978-3-319-27584-0

Mathematica®: A Problem-Centered Approach
Hazrat, R.
2015, XXI, 318 p. 164 illus., 139 illus. in color., Softcover
ISBN: 978-3-319-27584-0